

【ヒーターとサーモスタットによる水温調整】

- 風呂や電気ポットなどで水温(湯温)を設定温度に保つための制御機構(フィードバック制御)
- 実際には温度を測定してから温度調整までに時間がかかるので、その分の時間遅れをモデルに取り入れる
 - 変化後の水温 = 現在の水温 + 水温調整
 - 水温調整 = (設定温度 - 測定した水温) × ヒーター能力 × 時間間隔
 - 測定した水温 = (5 × 時間間隔) 前の水温
 - 設定温度: 40 度、ヒーター能力: 0.3、時間間隔: 0.5、最初の水温: 10 度

【Excel によるシミュレーション作成】

以下のような枠を作成し、数値を入力する(時間遅れを考慮して、その分のデータを先に入力)

	A	B	C	D	E	F	G	H
1								
2	時間	水温調整	水温		刻み幅	設定温度	ヒーター能力	
3	0	0	10		0.5	40	0.3	
4	0	0	10					
5	0	0	10					
6	0	0	10					
7	0	0	10					
8								
9								

時間、水温調整、水温の各数式を入力

- 時間 = 現在の時間 + 時間間隔
(A8セル) = A7 + \$E\$3
- 変化後の水温 = 現在の水温 + 水温調整
(C8セル) = C7 + B8
- 水温調整 = (設定温度 - 測定した水温) × ヒーター能力 × 時間間隔
※測定した水温 = (5 × 時間間隔) 前の水温
(B8セル) = (\$F\$3 - C3) * \$G\$3 * \$E\$3

	A	B	C	D
1				
2	時間	水温調整	水温	
3	0	0	10	
4	0	0	10	
5	0	0	10	
6	0	0	10	
7	0	0	10	
8	=A7+\$E\$3	=(F\$3-C3)*\$G\$3*\$E\$3	=C7+B8	
9				
10				

入力できたら下へコピー(時間=50 まで)し、「散布図」によるグラフを作成する

【複数の対象(例えばジュース缶)を温める場合】

下図のような枠を作成する(ここでは5本のジュースを1つのヒーターで温める)→ヒーター能力を本数で割って使用

	A	B	C	D	E	F	G	H	I	J	K
1	時間	温度1	温度2	温度3	温度4	温度5	本数		刻み幅	設定温度	ヒーター能力
2	0	30	30	30	30	30	①		1	80	0.5
3	0	30	30	30	30	30					
4	0	30	30	30	30	30					
5	0	30	30	30	30	30					
6	0	30	30	30	30	30					
7	1	②									
8	2										
9	3										


B2~F6セルに初期温度(ここでは30度)を入れておく

① (G2セル)=5-COUNTIF(B2:F2,0) ※ジュースが入っていない場合は0とする

② (B7セル)=B6+(\$J\$2-B2)*(\$K\$2/\$G2)*\$I\$2

①を下へ、②を右・下へコピー

ジュースの投入時刻をずらすなどとして試してみよう ※投入時刻の5ステップ前から温度欄に「30」と入れる



2種類の捕食者-被食者モデル

変化後のミジンコ個体数=現在のミジンコ個体数

$$+ (\text{ミジンコ増加率} \times \text{現在のミジンコ個体数})$$

$$- (\text{ミジンコ減少率} \times \text{現在のミジンコ個体数} \times \text{現在の熱帯魚個体数})$$

$$\times \text{時間間隔}$$

変化後の熱帯魚個体数=現在の熱帯魚個体数

$$+ (\text{熱帯魚増加率} \times \text{現在の熱帯魚個体数} \times \text{現在のミジンコ個体数})$$

$$- (\text{熱帯魚減少率} \times \text{現在の熱帯魚個体数})$$

$$\times \text{時間間隔}$$

	A	B	C	D	E	F	G	H	I
1	時間	ミジンコ個体数	熱帯魚個体数		時間間隔	ミジンコ減少率	ミジンコ増加率	熱帯魚減少率	熱帯魚増加率
2	0	500	10		0.05	0.15	1	0.25	0.0002
3	①	②	③						
4									
5									

① =A2+\$E\$2

② =B2+(\$G\$2*B2-\$F\$2*B2*C2)*\$E\$2

③ =C2+(\$I\$2*C2*B2-\$H\$2*C2)*\$E\$2

【数値例】

時間間隔 : 0.05

ミジンコ減少率 : 0.15


ミジンコ増加率 : 1

熱帯魚減少率 : 0.25

熱帯魚増加率 : 0.0002



ミジンコ個体数の初期値 : 500

熱帯魚個体数の初期値 : 10



ミジンコと熱帯魚の数がかなり違うので、熱帯魚のグラフを選んで「データ系列の書式設定」

→「軸」→「第2軸」とする



3種類の捕食者-被食者モデル

変化後の牧草個体数=現在の牧草個体数

$$+ (\text{牧草増加率} \times \text{現在の牧草個体数})$$

$$- (\text{牧草減少率} \times \text{現在の牧草個体数} \times \text{現在のヤギ個体数})$$

$$\times \text{時間間隔}$$

変化後のヤギ個体数=現在のヤギ個体数

$$+ (\text{ヤギ増加率} \times \text{現在のヤギ個体数} \times \text{現在の牧草個体数})$$

$$- (\text{ヤギ減少率} \times \text{現在のヤギ個体数} \times \text{現在のオオカミ個体数})$$

$$\times \text{時間間隔}$$

変化後のオオカミ個体数=現在のオオカミ個体数

$$+ (\text{オオカミ増加率} \times \text{現在のオオカミ個体数} \times \text{現在のヤギ個体数})$$

$$- (\text{オオカミ減少率} \times \text{現在のオオカミ個体数})$$

$$\times \text{時間間隔}$$

	A	B	C	D	E	F	G	H	I	J	K	L
1	時間	牧草個体数	ヤギ個体数	オオカミ個体数		時間間隔	牧草減少率	牧草増加率	ヤギ減少率	ヤギ増加率	オオカミ減少率	オオカミ増加率
2	0	500	10	5		0.05	0.15	1	0.25	0.0002	0.15	0.0225
3	①	②	③	④								
4												

- ① =A2+\$F\$2
- ② =B2+(\$H\$2*B2-\$G\$2*B2*C2)*\$F\$2
- ③ =C2+(\$J\$2*C2*B2-\$I\$2*C2*D2)*\$F\$2
- ④ =D2+(\$L\$2*D2*C2-\$K\$2*D2)*\$F\$2

【数値例】

時間間隔：0.05

牧草減少率：0.15

牧草増加率：0.9

ヤギ減少率：0.25

ヤギ増加率：0.005

オオカミ減少率：0.15


オオカミ増加率：0.0225

牧草個体数の初期値：200

ヤギ個体数の初期値：10

オオカミ個体数の初期値：2

それぞれ入力できたら下へコピーする。



ミジンコ-熱帯魚と同様に、「データ系列の書式設定」から第2軸を設定する。

牧草-ヤギ-オオカミの場合はヤギとオオカミのグラフを第2軸に合わせてやるとよい。

【ランダムウォークの例: 酔っ払いの歩行】

- 碁盤目状の道を酔っ払いがフラフラと歩いている。この酔っ払いは
 - ある地点から出発し、1回あたり上下左右いずれかに1/4の確率で進む
 - この酔っ払いはスタート地点からどのように進んで行くか

(1) 以下のような枠を作成する(第1行、A列の数字0~10が座標値となる)

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P	Q	R	S	T
1		0	1	2	3	4	5	6	7	8	9	10		乱数	x始点	y始点	x変化	y変化	x座標	y座標
2	0																			
3	1																			
4	2																			
5	3																			
6	4																			
7	5																			
8	6																			
9	7																			
10	8																			
11	9																			
12	10																			
13																				
14																				

(2) N2~T2セルに以下のように入力

	M	N	O	P	Q	R	S	T
1	乱数	x始点	y始点	x変化	y変化	x座標	y座標	
2	=RANDBETWEEN(1,4)	5	5	=IF(N2=1,Q2,IF(N2=2,Q2,IF(N2=3,Q2+1,Q2-1)))	=IF(N2=1,R2+1,IF(N2=2,R2-1,IF(N2=3,R2,R2)))	=O2+Q2	=P2+R2	

- (N2セル) =RANDBETWEEN(1,4) ... 1~4の1様乱数を生成
- (O2セル)、(P2セル) 5 ... スタート位置決定(座標5, 5)
- (Q2セル) =IF(N2=1,Q2,IF(N2=2,Q2,IF(N2=3,Q2+1,Q2-1))) ... 乱数が1、2のときはx座標はそのまま、3のときはx座標を1増加、それ以外(乱数が4)のときはx座標を1減少
- (R2セル) =IF(N2=1,R2+1,IF(N2=2,R2-1,IF(N2=3,R2,R2))) ... 乱数が1のときはy座標を1増加、2のときはy座標を1減少、乱数が3、4のときはy座標はそのまま
- (S2セル) =O2+Q2 ... x座標計算
- (T2セル) =P2+R2 ... y座標計算

※ 循環参照エラーが出るが、「キャンセル」を押してそのまま続行

(3) B3セルに以下のように入力

- (B3セル) =IF(AND(B\$1=\$S\$2,\$A2=\$T\$2),"●","")
 - ... B2セルの位置(B1、A2セルの番号)とS2、T2セルに書かれた座標位置が一致したら●、それ以外なら何も表示しない
- ※ できたら枠内へコピーする

(4) 循環参照の許可

- 循環参照を許可し、シミュレーションを実行可能にする
- 「ツール」→「オプション」→「計算方法」タブで計算方法を「手動」、「反復計算を許可」にチェックし、最大反復回数を「1」に
- できたらF9キーを押してシミュレーションを実行

壁ありランダムウォークの作成

(1) 以下のような枠を作成する（「W」は壁を表す）。

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P	Q	R	S
1	乱数	x方向	y方向	x座標	y座標		W	W	W	W	W	W	W	W	W	W	W	W	
2							W												W
3							W												W
4							W												W
5							W												W
6							W												W
7							W												W
8							W												W
9							W												W
10							W												W
11							W												W
12							W	W	W	W	W	W	W	W	W	W	W	W	W
13																			
14																			

(2) A2～E2 セルに以下のように入力する。

※循環参照エラーが出るがひとまずキャンセルを押して続行

A2セル： =RANDBETWEEN(1,4)

B2セル： =IF(AND(A2=2,INDIRECT(ADDRESS(E2+2,D2+8+1,4))<>"W"),1,
IF(AND(A2=4,INDIRECT(ADDRESS(E2+2,D2+8-1,4))<>"W"),-1,0))

C2セル： =IF(AND(A2=1,INDIRECT(ADDRESS(E2+2+1,D2+8,4))<>"W"),1,
IF(AND(A2=3,INDIRECT(ADDRESS(E2+2-1,D2+8,4))<>"W"),-1,0))

D2セル： =B2+D2

E2セル： =C2+E2

※比較演算子の「<>」は「等しくない（≠）」の意味。

内容の説明

A2セル：移動方向を決める乱数。ランダムに1～4を生成（1は下、2は右、3は上、4は左へ移動に対応、ただしその方向に壁があれば移動しない）。

B2セル：x方向の移動分を計算する。A2セルが2で、現在位置の右のセルが「W」でなければ1、A2セルが4で、現在位置の左のセルが「W」でなければ-1、それ以外は0。

C2セル：y方向の移動分を計算する。A2セルが1で、現在位置の下のセルが「W」でなければ1、A2セルが3で、現在位置の上のセルが「W」でなければ-1、それ以外は0。

D2セル：現在位置のx座標を計算。範囲は0～9になる。

E2セル：現在位置のy座標を計算。範囲は0～9になる。

ADDRESS関数：指定した行・列のセル番地を文字列で表示する関数。ADDRESS(y, x, 4)ならy行x列のセル番地を表示する。4は表示形式で、相対参照（通常のセル表示）となる

(1 とすれば絶対参照になる)。たとえば ADDRESS(3,2,4)は 3 行 2 列のセル番地「B3」となり、ADDRESS(1,5,1)なら 1 行 5 列のセル番地を絶対参照で「\$E\$1」となる。
このシミュレーションでは「現在位置 (の上下左右いずれか)」に相当するセル番地を取得するために使用している。ただし、実際に●が表示される領域は「2 行 8 列」にある H2 セルが原点(0, 0)になるため、x 座標に+8、y 座標に+2 としている。

INDIRECT 関数：文字列をセル参照の形式に変換できる関数。ADDRESS 関数で取得できるセル番地はあくまで「文字列」扱いなので、セル参照として扱うために INDIRECT 関数の中に入れる。

(3) H2 セルに以下のように入力し、W で囲まれた枠内 (H2～Q11 セル) にコピーする。

H2 セル： =IF(AND(COLUMN(H2)-8=\$D\$2,ROW(H2)-2=\$E\$2),"●","")

内容の説明

自分のセル番地における列番号-8が x 座標と、行番号-2が y 座標と等しければ「●」を表示し、そうでなければ何も表示しない。

COLUMN 関数：指定したセルの列番号を取得する関数。H2 セルなら 8 列目にあるので 8 となる。


ROW 関数：指定したセルの行番号を取得する関数。H2 セルなら 2 行目にあるので 2 となる。

※ここでは H2 セルが x,y 座標の原点(0, 0)になるため、それぞれ-8、-2 としている。


(4) H2 セルを Q11 セルまでコピーしたら、循環参照を許可し、F9 キーを押して動くことを確認する。確認できたら、枠内の好きなところに「W」を入力して壁を作成するとよい。

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P	Q	R	S
1	乱数	x方向	y方向	x座標	y座標		W	W	W	W	W	W	W	W	W	W	W	W	
2		4	-1	0	1	3	W												W
3							W												W
4							W				W								W
5							W	●			W								W
6							W				W								W
7							W			W	W	W			W				W
8							W				W				W	W	W		W
9							W									W			W
10							W									W			W
11							W												W
12							W	W	W	W	W	W	W	W	W	W	W	W	W
13																			
14																			

- ※ 「条件付き書式」を用いて「W」のセルを塗りつぶしてもよいと思います。
- ※ スタート位置は左上 (H2 セル) になるので、その付近には壁を作らないようにしてください。



【2重振り子のシミュレーション】



左図のように、重りが2つ繋がった振り子を考える。
支点の座標を(0, 0)とすると重りの座標は以下となる

$$\begin{cases} x_1 = L_1 \sin\theta_1 \\ y_1 = -L_1 \cos\theta_1 \\ x_2 = L_1 \sin\theta_1 + L_2 \sin\theta_2 \\ y_2 = -L_1 \cos\theta_1 - L_2 \cos\theta_2 \end{cases}$$

オイラー法を用いて近似すると、重りのなす角度は以下のように表現される

$$\begin{cases} \theta_1'(t + \Delta t) = \theta_1'(t) + \frac{A_2 D_1 - B D_2}{A_1 A_2 - B^2} \Delta t \\ \theta_2'(t + \Delta t) = \theta_2'(t) + \frac{A_1 D_2 - B D_1}{A_1 A_2 - B^2} \Delta t \\ \theta_1(t + \Delta t) = \theta_1(t) + \theta_1'(t) \Delta t \\ \theta_2(t + \Delta t) = \theta_2(t) + \theta_2'(t) \Delta t \end{cases}$$

ただし上式において

$$\begin{cases} A_1 = (M_1 + M_2)L_1^2 \\ A_2 = M_2 L_2^2 \\ B = M_2 L_1 L_2 \cos(\theta_1 - \theta_2) \\ D_1 = -M_2 L_1 L_2 \theta_2' \sin(\theta_1 - \theta_2) - (M_1 + M_2) g L_1 \sin\theta_1 \\ D_2 = M_2 L_1 L_2 \theta_1' \sin(\theta_1 - \theta_2) - M_2 g L_2 \sin\theta_2 \end{cases}$$

以上を用いて Excel のシミュレーションを作成する

(1) パラメータの入力

以下のように必要な数値を入力する

	A	B	C	D	E	F	G
1	M1	M2	L1	L2	$\theta_1(0)$	$\theta_2(0)$	Δt
2	1.2	0.1	1.4	1.2	2.5	2.8	0.05
3							

(2) 時刻の入力

計算に必要な時刻を入力（今回は30でリセットされるようしておく）

※循環参照の警告が出るが、いったん無視して閉じる

	A	B	C	D	E	F	G
1	M1	M2	L1	L2	$\theta_1(0)$	$\theta_2(0)$	Δt
2	1.2	0.1	1.4	1.2	2.5	2.8	0.05
3	$t + \Delta t$	t					
4							
5							
6							
7							
8							
9							

セル D5 に `=A4-G2` と入力
セル A8 に `=IF(A4>=30,G2,A4+G2)` と入力

(3) 角度計算に必要な数値 A1~D2 の計算を行う

	A	B	C	D	E	F	G	H	I
1	M1	M2	L1	L2	$\theta 1(0)$	$\theta 2(0)$	Δt		
2	1.2	0.1	1.4	1.2	2.5	2.8	0.05		
3	$t + \Delta t$	t							
4	0.05	0							
5									
6	A1	A2	B	D1	D2				
7									
8									
9		$=B2*D2*D2$			$=B2*C2*D2*A9*SIN(A11-B11)-B2*9.8*D2*SIN(B11)$				
10	$=(A2+B2)*C2*C2$			$=-1*B2*C2*D2*B9*SIN(A11-B11)-(A2+B2)*9.8*C2*SIN(A11)$					
11			$=B2*C2*D2*COS(A11-B11)$						
12									
13									
14									

(4) 角度および角度の微分（角速度）を計算する

	A	B	C	D	E	F	G	H
1	M1	M2	L1	L2	$\theta 1(0)$	$\theta 2(0)$	Δt	
2	1.2	0.1	1.4	1.2	2.5	2.8	0.05	
3	$t + \Delta t$	t						
4	25.75	25.7						
5								
6	A1	A2	B	D1	D2			
7	2.548	0.144	0.168	0	0			
8	$\theta 1'$	$\theta 2'$						
9								
10	$\theta 1$	$\theta 2$						
11								
12								
13								
14	$=IF(B4=0,E2,A11+A9*G2)$							
15								
16								


(5) 重りの座標を計算する

※グラフ作成の都合上、支点の座標を(0,0)とする

	A	B	C	D	E	F	G
4	2.55	2.5					
5							
6	A1	A2	B	D1	D2		
7	2.548	0.144	0.0630044	16.686922	0.4674984		
8	$\theta 1'$	$\theta 2'$					
9	2.7756784	1.8996529					
10	$\theta 1$	$\theta 2$					
11	-1.743123	9.5930757					
12							
13		x	y				
14	支点	0	0				
15	1						
16	2						
17							
18	$=B15+D2*SIN(B11)$		$=C15-D2*COS(B11)$				
19							


(6) 循環参照の許可

数式が入力できたら、循環参照を許可して繰り返し計算ができるようにする
「ファイル」メニューから「オプション」を選び、「数式」タブから計算方法を「手動」、
「反復計算を許可する」にチェックを入れて「最大反復回数」を「1」にする




(7) グラフの作成

x、y 座標を選択し、「挿入」メニューから「散布図（マーカーと直線）」を選択する



(8) グラフの調整

x、y 軸を選択し、座標範囲を-3～+3にする
線の色、マーカーの色や大きさなどを変更し、振り子らしく見えるようにする
(右クリック⇒データ系列の書式設定 など)



※完成したら F9 キーで動作するか確かめよう

【入力する計算式のまとめ】

	A	B	C	D	E	F	G
1	M1	M2	L1	L2	$\theta 1(0)$	$\theta 2(0)$	Δt
2	1.2	0.1	1.4	1.2	2.5	2.8	0.05
3	$t + \Delta t$	t					
4	①	②					
5							
6	A1	A2	B	D1	D2		
7	③	④	⑤	⑥	⑦		
8	$\theta 1'$	$\theta 2'$					
9	⑧	⑨					
10	$\theta 1$	$\theta 2$					
11	⑩	⑪					
12							
13		x	y				
14	支点	0	0				
15	1	⑫	⑬				
16	2	⑭	⑮				
17							

① =IF(A4>=30,G2,A4+G2)

② =A4 - G2

③ =(A2+B2)*C2*C2

④ =B2*D2*D2

⑤ =B2*C2*D2*COS(A11 - B11)

⑥ =-1*B2*C2*D2*B9*SIN(A11 - B11) - (A2+B2)*9.8*C2*SIN(A11)

⑦ =B2*C2*D2*A9*SIN(A11 - B11) - B2*9.8*D2*SIN(B11)

⑧ =IF(B4=0,0,A9+G2*(B7*D7 - C7*E7)/(A7*B7 - C7*C7))

⑨ =IF(B4=0,0,B9+G2*(A7*E7 - C7*D7)/(A7*B7 - C7*C7))

⑩ =IF(B4=0,E2,A11+A9*G2)

⑪ =IF(B4=0,F2,B11+B9*G2)

⑫ =C2*SIN(A11)


⑬ =-1*C2*COS(A11)

⑭ =B15+D2*SIN(B11)

⑮ =C15 - D2*COS(B11)

【強化学習による迷路探索】

●で表されるエージェントが迷路内を移動し、赤色で表されたゴールへ向かう道順を学習するシミュレーションを作成する



(1) 枠の作成

A列以外をほぼ正方形になるよう調整しておく

まず B1~H8 セルを囲い、一番外側（壁になる部分）に「-1」、G3セル（ゴール）に「1」を入力

(2) Q値の計算枠を作成

10~15行目を使い、学習のためのQ値を入力する枠を作る

マイナスの指定ができるよう、x、y座標も書き込んでおく

※ y座標は上から下に向かって増加するようにする

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P	Q	R	S	T	U	V	W	X	Y	Z	AA	AB	AC
1																													
2		-1	-1	-1	-1	-1	-1	-1																					
3		-1	0	0	0	0	0	1	-1																				
4		-1	0	0	0	0	0	0	-1																				
5		-1	0	0	0	0	0	0	-1																				
6		-1	0	0	0	0	0	0	-1																				
7		-1	0	0	0	0	0	0	-1																				
8		-1	-1	-1	-1	-1	-1	-1	-1																				
9																													
10			0	1	2	3	4			0	1	2	3	4			0	1	2	3	4			0	1	2	3	4	
11		0																											
12		1																											
13		2																											
14		3																											
15		4																											

(3) 現在地のQ値の計算

17~18行目に以下のように入力する

17行目は●のいる場所の4つのQ値（上、右、下、左）となる

18行目は、Q値の最大値が複数ある場合、この列の値（ランダム）を使って1つを選択する

16																														
17	Q(t)	1	1	1	1																									
18	ランダム数	0.41	0.83	0.69	0.51																									

(B17セル) =INDIRECT(ADDRESS(B28+11,B27+3))
 (C17セル) =INDIRECT(ADDRESS(B28+11,B27+10))
 (D17セル) =INDIRECT(ADDRESS(B28+11,B27+17))
 (E17セル) =INDIRECT(ADDRESS(B28+11,B27+24))

※C18~E18はB18を横へドラッグしてコピー

(B18セル) =IF(B17=MAX(\$B\$17:\$E\$17),RAND(),-1)

(4) 学習のための各種要素を計算

20～31 行目に以下のような数式を入力する

	A	B	C	D	E	F
19						
20	t	⑩				
21	T	⑪				
22	行動	⑫				
23	x(t)	⑬				
24	y1(t)	⑭				
25	dx	⑮				
26	dy	⑯				
27	x(t+1)	⑰				
28	y(t+1)	⑱				
29	報酬	⑲				
30	Q(t+1)	⑳	㉑	㉒	㉓	
31	Qmax	㉔				

⑩ (B20 セル) =IF(B29=1,0,IF(B20<100,B20+1,0))

⑪ (B21 セル) =IF(B20=0,B21+1,B21)

⑫ (B22 セル) =IF(B18=MAX(B18:E18),1,
IF(C18=MAX(B18:E18),2,
IF(D18=MAX(B18:E18),3,4)))

⑬ (B23 セル) =IF(B20=0,0,B27)

⑭ (B24 セル) =IF(B20=0,4,B28)

⑮ (B25 セル) =IF(B22=2,1,IF(B22=4,-1,0))

⑯ (B26 セル) =IF(B22=1,-1,IF(B22=3,1,0))

⑰ (B27 セル) =IF(INDIRECT(ADDRESS(B24+B26+3,B23+B25+3))=-1,
B23,B23+B25)

⑱ (B28 セル) =IF(INDIRECT(ADDRESS(B24+B26+3,B23+B25+3))=-1,
B24,B24+B26)

⑲ (B29 セル) =INDIRECT(ADDRESS(B24+B26+3,B23+B25+3))

⑳ (B30 セル) =INDIRECT(ADDRESS(B28+11,B27+3))

㉑ (C30 セル) =INDIRECT(ADDRESS(B28+11,B27+10))

㉒ (D30 セル) =INDIRECT(ADDRESS(B28+11,B27+17))

㉓ (E30 セル) =INDIRECT(ADDRESS(B28+11,B27+24))

㉔ (B31 セル) =MAX(B30:E30)

※循環参照の警告はとりあえず無視する

(5) 迷路の状態を記述

迷路とエージェントを表示する枠を作る

C34~I40 セルを囲い、外側に x、y 座標を記入しておく

左上の C34 セルに以下の数式を入力し、枠内にコピーする

	A	B	C	D	E	F	G	H	I	J
31	Qmax	1								
32										
33			-1	0	1	2	3	4	5	
34		-1								
35		0								
36		1								
37		2								
38		3								
39		4								
40		5								
41										

(C34セル) =IF(AND(C\$33=\$B\$27,\$B34=\$B\$28),"●",B2)

(6) Q 値の計算

C11、J11、Q11、X11 の各セルに Q 値を計算する式を入力し、それぞれの枠内にドラッグしてコピーする

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P	Q	R	S	T	U	V	W	X	Y	Z	AA	AB	AC
10			0	1	2	3	4			0	1	2	3	4			0	1	2	3	4			0	1	2	3	4	
11		0	0	1	1	1	1	1	0	1	1	1	1	1	0	1	1	1	1	1	1	0	1	1	1	1	1	1	
12		1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	
13		2	1	1	1	1	1	1	2	1	1	1	1	1	1	1	2	1	1	1	1	1	1	2	1	1	1	1	
14		3	1	1	1	1	1	1	3	1	1	1	1	1	1	3	1	1	1	1	1	1	3	1	1	1	1	1	
15		4	1	1	1	1	1	1	4	1	1	1	1	1	1	4	1	1	1	1	1	1	4	1	1	1	1	1	
16																													

(C11セル) =IF(AND(\$B\$20=0,\$B\$21=0),1,IF(AND(\$B\$23=C\$10,\$B\$24=\$B11,\$B\$22=1),0.5*C11+0.5*(\$B\$29+0.9*\$B\$31),C11))

(J11セル) =IF(AND(\$B\$20=0,\$B\$21=0),1,IF(AND(\$B\$23=J\$10,\$B\$24=\$B11,\$B\$22=2),0.5*J11+0.5*(\$B\$29+0.9*\$B\$31),J11))


(Q11セル) =IF(AND(\$B\$20=0,\$B\$21=0),1,IF(AND(\$B\$23=Q\$10,\$B\$24=\$B11,\$B\$22=3),0.5*Q11+0.5*(\$B\$29+0.9*\$B\$31),Q11))

(X11セル) =IF(AND(\$B\$20=0,\$B\$21=0),1,IF(AND(\$B\$23=X\$10,\$B\$24=\$B11,\$B\$22=4),0.5*X11+0.5*(\$B\$29+0.9*\$B\$31),X11))

(7) 循環参照の許可

数式が入力できたら、循環参照を許可して繰り返し計算ができるようにする

「ファイル」メニューから「オプション」を選び、「数式」タブから計算方法を「手動」、
「反復計算を許可する」にチェックを入れて「最大反復回数」を「1」にする



(8) 枠内に色を付ける

「条件付き書式」により、迷路の枠内に色をつける

「条件付き書式」 → 「セルの強調表示ルール」 → 「指定の値に等しい」として、「-1」の時は黒、「0」のときは白、「1」のときは赤になるようにする

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P	Q	R	S	T	U	V	
22	行動	2																					
23	x(t)	1																					
24	y1(t)	0																					
25	dx	1																					
26	dy	0																					
27	x(t+1)	2																					
28	y(t+1)	0																					
29	報酬	0																					
30	Q(t+1)	1	1	1	1																		
31	Qmax	1																					
32																							
33			-1	0	1	2	3	4	5														
34		-1																					
35		0		●					1														
36		1																					
37		2																					
38		3																					
39		4																					
40		5																					
41																							

指定の値に等しい

次の値に等しいセルを書式設定:

-1

書式: ユーザー設定の書式...

OK キャンセル

(9) 完成

F9 キーを押して動作させ、ゴールへの経路を学習するか確認

2~8行目に作った枠内に「-1」を入力すると、その場所が壁になるので、自由に迷路を作って学習させてみよう

	A	B	C	D	E	F	G	H	I
1									
2		-1	-1	-1	-1	-1	-1	-1	
3		-1	0	0	0	0	1	-1	
4		-1	0	0	0	0	0	-1	
5		-1	0	-1	-1	-1	0	-1	
6		-1	0	0	0	0	0	-1	
7		-1	0	0	0	0	0	-1	
8		-1	-1	-1	-1	-1	-1	-1	